The goal is to make a program that finds conditions of equilibrium between two polymer phases and plot it.  The end result is this plot:




Start by drawing naming the form frmFH.

Add these **label** elements:

Label1 (I should really have named it something like lblTitle)

lblSigma lblChiCrit lblCrit lblEqual

lblPhiCrit lblCrit2 lblEqual2

lblPhiX lblSub2 lblChiAxis

Note:  I used different labels to make the subscripts, equal signs, etc.

Add in the 3 **text boxes**:  txtSigma, txtChiCrit, txtPhi2Crit

Add the **buttons**: cmdGo and cmdStop, and give them captions &Compute and &Stop

Add an **invisible label**: lblComputing (you don't see this in the image above, but we'll make it appear during the calculations). Set its caption to "Computing!" Give it a yellow background and red font. Set its "Visible" property to "False".

Double click the Stop button and you'll see an empty subroutine. Make it like this:

```
Private Sub cmdStop_Click()
End
End Sub
```

Double click the Compute Button and you'll see an empty subroutine. Make it like this:

```
Private Sub cmdGo_Click()
  lblComputing.Visible = True
  MainStuff
End Sub
```

Go to "Project" menu. Add a module and immediately name it FHModule.

Now is a good time to save everything! Be sure to make a NEW DIRECTORY to hold the saved project, as this makes it easy to keep track of everything.

Into your new FHModule, type the following code:

```
Option Explicit
Public ASTORE, BSTORE, CHISTORE   'arrays to store A, B, Chi Values
Public Sigma  'degree of polymerization
Public ChiCrit, Phi2Crit  'critical Chi and Phi2 values (analytically determined from
sigma)

Sub MainStuff()
'ADOPTED FROM QUICK BASIC PROGRAM
'COMPUTES FLORY-HUGGINS PHASE DIAGRAMS
'IDEA IS TO FIND PHI2CRIT AND CHICRIT ANALYTICALLY.  THEN
'SELECT A PHI2B > PHI2CRIT AND GO SEARCHING FOR PHI2A AND
'THE CORRESPONDING CHI.  THE SEARCH IS TO MAKE TWO EQUATIONS
'DERIVED FOR CHI AND STEMMING FROM THE EQUILIBRIUM CONDITIONS
(MU1A = MU1B
'AND MU2A = MU2B) SIMULTANEOUSLY YIELD THE SAME VALUE OF CHI.
'
'THESE TWO EQUATIONS ARE 'A' AND 'B'.

'THE PROGRAM INCLUDES A VERY CRUDE PLOTTING ROUTINE.  IF YOU
HAD TO DO
```

```
'THESE THINGS ALL DAY LONG, YOU MIGHT FIND THIS PROGRAM FASTER
THAN
'MATHCAD OR SIMILAR.  HOWEVER, YOU WOULD DEFINITELY WISH TO
MAKE THE
'SEARCH MORE INTELLIGENT.
'
ReDim ASTORE(100), BSTORE(100), CHISTORE(100)
Dim Astep, Bstep  'step size increments
Dim NB, NA  ' loop indices
Dim ChiCrit, Phi2Crit   'critical points
Dim AStart  'Start Searching goint left from ctitical point; changes when solution found
Dim Phi2B  'Phi2 beta phase
Dim Phi2A  'Phi2 alpha phase
Dim LowDiff 'difference finding variable; initially set to something huge...goes down
each time better solution found
Dim ChiE, ChiF  'Chi parameters from Equations E and F in handwritten solution
Dim Diff 'absolute value of difference (ChiE-ChiF)
Dim BestA 'best alpha value for a given beta
Dim BestChi 'corresponding Chi value
Dim msg 'message variable to appear in msgbox

'MsgBox "In MainStuff"
'frmFH.picPhase.Cls
msg = ""  'start with blank message
Sigma = Val(frmFH.txtSigma.Text)
ChiCrit = (1 + Sqr(Sigma)) ^ 2 / (2 * Sigma)
Phi2Crit = 1 / (1 + Sqr(Sigma))
frmFH.txtChiCrit.Text = ChiCrit
frmFH.txtPhi2Crit.Text = Phi2Crit
frmFH.picPhase.Scale (0, 1)-(1, 0.5)
frmFH.txtYmaxLegend.Text = 1
frmFH.txtYminLegend.Text = 0.5

DoEvents  'causes VB to update screens, etc.

'Print "SIGMA:  "; Sigma; "  CHI CRITICAL:  "; ChiCrit; "   PHI2 CRITICAL:  ";
Phi2Crit

Bstep = 0.008  'LOOK FOR PHI2B POINTS AT PHI2CRIT + N*Bstep WHERE N =
INTEGER
AStart = Phi2Crit 'START SEARCHING "GOING LEFT" FROM CRIT POINT.
                    'THIS WILL BE CHANGED AFTER A SOLUTION HAS BEEN
FOUND.
Astep = Bstep / 100    'MY COMPUTER IS FAIRLY FAST, SO I CAN SEARCH
                    'CAREFULLY USING A SMALL STEP.
For NB = 1 To 50
```

```
  Phi2B = Phi2Crit + NB * Bstep
  LowDiff = 100    'SET THE DIFFERENCE FINDER TO SOMETHING HUGE
  For NA = 1 To 10000
  Phi2A = AStart - Astep * NA
    If Phi2A <= 0 Then
      GoTo ENDA   'END LOOP WHEN PHI2A GOES NEGATIVE & STORE BEST
VALUES
    End If

    ChiE = (Log((1 - Phi2B) / (1 - Phi2A)) + (1 - 1 / Sigma) * (Phi2B - Phi2A)) / (Phi2A ^
2 - Phi2B ^ 2)
    ChiF = (Log(Phi2B / Phi2A) + (1 - Sigma) * (Phi2A - Phi2B)) / (Sigma * ((1 - Phi2A)
^ 2 - (1 - Phi2B) ^ 2))

    Diff = Abs(ChiE - ChiF)
    If Diff < LowDiff Then
        LowDiff = Diff
        BestA = Phi2A
        BestChi = ChiE
    End If
  Next NA
ENDA:
  ASTORE(NB) = BestA
  BSTORE(NB) = Phi2B
  CHISTORE(NB) = BestChi
  msg = msg & BSTORE(NB) & " " & ASTORE(NB) & " " & CHISTORE(NB) &
vbCrLf   'add the results to msg
  frmFH.picPhase.Circle (BestA, BestChi), 0.002  'plot the best alpha point found
  frmFH.picPhase.Circle (Phi2B, BestChi), 0.002  'plot the beta point
  AStart = BestA  'now we will only look to the left of our previous alpha point
Next NB

frmFH.picPhase.PSet (Phi2Crit, ChiCrit) 'PLOT THE CRITICAL POINT
'MsgBox msg
frmFH.lblComputing.Visible = False
End Sub
```

APPENDIX. Theory of what we're doing here. The conditions $\mu_1^\alpha=\mu_1^\beta$ and $\mu_2^\alpha=\mu_2^\beta$ lead to two different equations whose simultaneous solution (i.e., $\phi_2^\alpha$ and $\phi_2^\beta$) is sought. Each equation is solved analytically for $\chi$, which must be the same in each phase (because $\chi$ is effectively an inverse temperature). So then you equate the two equations for $\chi$. Pick some value of $\phi_2^\beta$ (e.g., to the left of the critical point, which you compute analytically). Search by trial and error for $\phi_2^\alpha$ values that make the left-hand side of the equation and right hand side equal. Choose tight tolerances (i.e., demand they be really equal).

Here it is in equations (hand-written). ChiE and ChiF in the VB code are called ChiA and ChiB below.

Equilibrium occurs when $\mu_1(\phi_2^\alpha) = \mu_1(\phi_2^\beta)$ AND $\mu_2(\phi_2^\alpha) = \mu_2(\phi_2^\beta)$

from previous problem we get

$$\ln(1-\phi_2^\alpha) + (1-\phi_2^\alpha)\left(1-\frac{1}{\sigma}\right) + \chi\phi_2^{\alpha 2} = \ln(1-\phi_2^\beta) + (1-\phi_2^\beta)\left(1-\frac{1}{\sigma}\right) + \chi\phi_2^{\beta 2} \qquad (A)$$

AND

$$\ln\phi_2^\alpha + (1-\sigma)(1-\phi_2^\alpha) - \chi\sigma(1-\phi_2^\alpha)^2 = \ln\phi_2^\beta + (1-\sigma)(1-\phi_2^\beta) - \chi\sigma(1-\phi_2^\beta)^2 \qquad (B)$$

Solve (A) for $\chi$. Call it

$$\chi_A = \frac{\ln\left(\frac{1-\phi_2^\beta}{1-\phi_2^\alpha}\right) + \left(1-\frac{1}{\sigma}\right)(\phi_2^\beta - \phi_2^\alpha)}{\phi_2^{\alpha 2} - \phi_2^{\beta 2}}$$

Solve (B) for $\chi$. Call it

$$\chi_B = \frac{\ln(\phi_2^\beta/\phi_2^\alpha) + (1-\sigma)[\phi_2^\alpha - \phi_2^\beta]}{\sigma\left[(1-\phi_2^\alpha)^2 - (1-\phi_2^\beta)^2\right]}$$

Since $\chi = \frac{Z\epsilon}{kT}$ it must be the same in both phases, since temp must be same.

Strategy: Set 1) $\chi_A = \chi_B$

2) Choose $\phi_2^\beta$

3) Hunt for $\phi_2^\alpha$ that make it work.

Before starting, consider where critical points are.

$$\chi_{crit} = \frac{(1+\sqrt{\sigma})^2}{2\sigma} \qquad \phi_{2,crit} = \frac{1}{1+\sqrt{\sigma}}$$

@ $\sigma = 65$, $\chi_{crit} = 0.632$ and $\phi_{2,crit} = 0.11$

The Basic Program provided solves the $\chi_A = \chi_B$ for progressively larger values of $\phi_2^\beta$. It also makes a crude plot.

A MATH CAD page and EXCEL spreadsheet appear also. There are many alternative approaches.